Rules for integrands of the form $(a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n$

1. $\int (a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n \, dx$ when $b c + a d == 0 \wedge a^2 - b^2 == 0$

1: $\int (a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n \, dx$ when $b c + a d == 0 \wedge a^2 - b^2 == 0 \wedge m \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Rule: If $b c + a d == 0 \wedge a^2 - b^2 == 0 \wedge m \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^-$, then

$$\int (a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n \, dx \rightarrow c^n \int \left(1 + \frac{d}{c} Sec[e + f x]\right)^n ExpandTrig\left[(a + b \, Sec[e + f x])^m, x\right] dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  c^n*Int[ExpandTrig[(1+d/c*csc[e+f*x])^n,(a+b*csc[e+f*x])^m,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IGtQ[m,0] && ILtQ[n,0] && LtQ[m+n,2]
```

2: $\int (a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n \, dx$ when $b c + a d == 0 \wedge a^2 - b^2 == 0 \wedge m \in \mathbb{Z} \wedge n \in \mathbb{R}$

Derivation: Algebraic simplification

Basis: If $b c + a d == 0 \wedge a^2 - b^2 == 0$, then $(a + b \, Sec[z]) (c + d \, Sec[z]) == -a c \, Tan[z]^2$

Rule: If $b c + a d == 0 \wedge a^2 - b^2 == 0 \wedge m \in \mathbb{Z} \wedge n \in \mathbb{R}$, then

$$\int (a + b \, Sec[e + f x])^m (c + d \, Sec[e + f x])^n \, dx \rightarrow (-a c)^m \int Tan[e + f x]^{2m} (c + d \, Sec[e + f x])^{n-m} \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  (-a*c)^m*Int[Cot[e+f*x]^(2*m)*(c+d*Csc[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] && RationalQ[n] && Not[IntegerQ[n] && GtQ[m-n,0]]
```

3: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^m\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{1}{2}\in\mathbb{Z}$, then $(a+b\,\text{Sec}[z])^m\,(c+d\,\text{Sec}[z])^m==\frac{(-a\,c)^{m+\frac{1}{2}}\,\text{Tan}[z]^{2\,m+1}}{\sqrt{a+b\,\text{Sec}[z]}\,\sqrt{c+d\,\text{Sec}[z]}}$

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==0$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m+\frac{1}{2}\in\mathbb{Z}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^m\,dx\;\rightarrow\;\frac{(-a\,c)^{m+\frac{1}{2}}\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\int \text{Tan}[e+f\,x]^{2\,m}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  (-a*c)^(m+1/2)*Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Int[Cot[e+f*x]^(2*m),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m+1/2]
```

4. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\land a^2-b^2==0$

1: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\land a^2-b^2==0\land n>\frac{1}{2}$

Rule: If $b\,c+a\,d==0\land a^2-b^2==0\land n>\frac{1}{2}$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\longrightarrow$$

$$-\frac{2\,a\,c\,\text{Tan}[e+f\,x]\,(c+d\,\text{Sec}[e+f\,x])^{n-1}}{f\,(2\,n-1)\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}+c\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  2*a*c*Cot[e+f*x]*(c+d*Csc[e+f*x])^(n-1)/(f*(2*n-1)*Sqrt[a+b*Csc[e+f*x]]) +
  c*Int[Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && GtQ[n,1/2]
```

2: $\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (c + d \operatorname{Sec}[e + f x])^n \, dx$ when $b c + a d == 0 \land a^2 - b^2 == 0 \land n < -\frac{1}{2}$

Rule: If $b c + a d == 0 \land a^2 - b^2 == 0 \land n < -\frac{1}{2}$, then

$$\int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (c + d \operatorname{Sec}[e + f x])^n \, dx \longrightarrow$$

$$\frac{2 a \operatorname{Tan}[e + f x] \, (c + d \operatorname{Sec}[e + f x])^n}{f (2 n + 1) \sqrt{a + b \operatorname{Sec}[e + f x]}} + \frac{1}{c} \int \sqrt{a + b \operatorname{Sec}[e + f x]} \, (c + d \operatorname{Sec}[e + f x])^{n+1} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -2*a*Cot[e+f*x]*(c+d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
  1/c*Int[Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && LtQ[n,-1/2]
```

5. $\int (a+b \operatorname{Sec}[e+f x])^{3/2} (c+d \operatorname{Sec}[e+f x])^n \, dx$ when $b c+a d==0 \land a^2-b^2==0$

1: $\int (a+b \operatorname{Sec}[e+f x])^{3/2} (c+d \operatorname{Sec}[e+f x])^n \, dx$ when $b c+a d==0 \land a^2-b^2==0 \land n<-\frac{1}{2}$

Rule: If $b c+a d==0 \land a^2-b^2==0 \land n<-\frac{1}{2}$, then

$$\int (a+b \operatorname{Sec}[e+f x])^{3/2} (c+d \operatorname{Sec}[e+f x])^n \, dx \longrightarrow$$

$$\frac{4 a^2 \operatorname{Tan}[e+f x] (c+d \operatorname{Sec}[e+f x])^n}{f (2 n+1) \sqrt{a+b \operatorname{Sec}[e+f x]}}+\frac{a}{c} \int \sqrt{a+b \operatorname{Sec}[e+f x]} \, (c+d \operatorname{Sec}[e+f x])^{n+1} \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  -4*a^2*Cot[e+f*x]*(c+d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
  a/c*Int[Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && LtQ[n,-1/2]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^{3/2}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d=0\,\land\,a^2-b^2=0\,\land\,n\nleq-\frac{1}{2}$

Rule: If $b\,c+a\,d=0\,\land\,a^2-b^2=0\,\land\,n\nleq-\frac{1}{2}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^{3/2}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{2\,a^2\,\text{Tan}[e+f\,x]\,(c+d\,\text{Sec}[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}+a\int\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  -2*a^2*Cot[e+f*x]*(c+d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
  a*Int[Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LeQ[n,-1/2]]
```

6: $\int (a+b\,\text{Sec}[e+f\,x])^{5/2}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,n<-\frac{1}{2}$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,n<-\frac{1}{2}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^{5/2}\,(c+d\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow$$
$$\frac{8\,a^3\,\text{Tan}[e+f\,x]\,(c+d\,\text{Sec}[e+f\,x])^n}{f\,(2\,n+1)\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}+\frac{a^2}{c^2}\int\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])^{n+2}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(5/2)*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  -8*a^3*Cot[e+f*x]*(c+d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
  a^2/c^2*Int[Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])^(n+2),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && LtQ[n,-1/2]
```

7: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}\,\wedge\,m+n==0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==0$

Basis: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0$, then $-\frac{a\,c\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==1$

Basis: $\text{Tan}[e+f\,x]\,F[\text{Sec}[e+f\,x]]==-\frac{1}{f}\,\text{Subst}\left[\frac{F\left[\frac{1}{x}\right]}{x},\,x,\,\text{Cos}[e+f\,x]\right]\,\partial_x\text{Cos}[e+f\,x]$

Rule: If $b\,c+a\,d==0\,\wedge\,a^2-b^2==0\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}\,\wedge\,m+n==0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx \longrightarrow -\frac{a\,c\,\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\int \text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\text{Sec}[e+f\,x])^{n-\frac{1}{2}}\,dx$$

$$\to \frac{a\,c\,\text{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,\text{Subst}\Big[\int\frac{(b+a\,x)^{m-\frac{1}{2}}\,(d+c\,x)^{n-\frac{1}{2}}}{x^{m+n}}\,dx,\,x,\,\text{Cos}[e+f\,x]\Big]$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  -a*c*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*
    Subst[Int[(b+a*x)^(m-1/2)*(d+c*x)^(n-1/2)/x^(m+n),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m-1/2] && EqQ[m+n,0]
```

8: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $\partial_x\frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}==0$

Basis: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then $-\frac{a\,c\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\,\frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}==1$

Basis: $\mathrm{Tan}[e+f\,x]\,F[\mathrm{Sec}[e+f\,x]]==\frac{1}{f}\,\mathrm{Subst}\left[\frac{F[x]}{x},\,x,\,\mathrm{Sec}[e+f\,x]\right]\,\partial_x\mathrm{Sec}[e+f\,x]$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\rightarrow\,-\frac{a\,c\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\int \mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sec}[e+f\,x])^{n-\frac{1}{2}}\,dx$$

$$\rightarrow\,-\frac{a\,c\,\mathrm{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\,\mathrm{Subst}\left[\int\frac{(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^{n-\frac{1}{2}}}{x}\,dx,\,x,\,\mathrm{Sec}[e+f\,x]\right]$$

–

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  a*c*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^(n-1/2)/x,x],x,Csc[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0]
```

2. $\int (a + b \operatorname{Sec}[e + f x])^m (c + d \operatorname{Sec}[e + f x]) \, dx$ when $b c - a d \neq 0$

   1. $\int (a + b \operatorname{Sec}[e + f x])^m (c + d \operatorname{Sec}[e + f x]) \, dx$ when $b c - a d \neq 0 \land m > 0$

      1. $\int (a + b \operatorname{Sec}[e + f x]) (c + d \operatorname{Sec}[e + f x]) \, dx$ when $b c - a d \neq 0$

         1: $\int (a + b \operatorname{Sec}[e + f x]) (c + d \operatorname{Sec}[e + f x]) \, dx$ when $b c + a d == 0$

Derivation: Algebraic expansion

Basis: If $b c + a d == 0$, then $(a + b z) (c + d z) == a c + b d z^2$

Rule: If $b c + a d == 0$, then

$$\int (a + b \operatorname{Sec}[e + f x]) (c + d \operatorname{Sec}[e + f x]) \, dx \longrightarrow a c x + b d \int \operatorname{Sec}[e + f x]^2 \, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*c*x + b*d*Int[Csc[e+f*x]^2,x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[b*c+a*d,0]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\land b\,c+a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $(c+d\,z)\,(a+b\,z)=a\,c+(b\,c+a\,d)\,z+b\,d\,z^2$

Rule: If $b\,c-a\,d\neq 0\land b\,c+a\,d\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])\,(c+d\,\text{Sec}[e+f\,x])\,dx\longrightarrow a\,c\,x+(b\,c+a\,d)\int \text{Sec}[e+f\,x]\,dx+b\,d\int \text{Sec}[e+f\,x]^2\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*c*x + (b*c+a*d)*Int[Csc[e+f*x],x] + b*d*Int[Csc[e+f*x]^2,x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[b*c+a*d,0]
```

2. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0$

Derivation: Algebraic expansion

Rule: If $b\,c-a\,d\neq 0\wedge a^2-b^2==0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])\,dx\rightarrow c\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx+d\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\text{Sec}[e+f\,x]\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  c*Int[Sqrt[a+b*Csc[e+f*x]],x] + d*Int[Sqrt[a+b*Csc[e+f*x]]*Csc[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{a+b\,z}\,(c+d\,z)=\frac{a\,c}{\sqrt{a+b\,z}}+\frac{z\,(b\,c+a\,d+b\,d\,z)}{\sqrt{a+b\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])\,dx\,\rightarrow\,a\,c\int\frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+\int\frac{\text{Sec}[e+f\,x]\,(b\,c+a\,d+b\,d\,\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*c*Int[1/Sqrt[a+b*Csc[e+f*x]],x] +
  Int[Csc[e+f*x]*(b*c+a*d+b*d*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

3. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\land m>1$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\land m>1\land a^2-b^2==0$

Derivation: Singly degenerate secant recurrence 1b with $n\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\land m>1\land a^2-b^2==0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx\longrightarrow$$

$$\frac{b\,d\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-1}}{f\,m}+\frac{1}{m}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m-1}\,(a\,c\,m+(b\,c\,m+a\,d\,(2\,m-1))\,\mathrm{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)/(f*m) +
  1/m*Int[(a+b*Csc[e+f*x])^(m-1)*Simp[a*c*m+(b*c*m+a*d*(2*m-1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && GtQ[m,1] && EqQ[a^2-b^2,0] && IntegerQ[2*m]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,m>1\,\land\,a^2-b^2\neq 0$

Derivation: Cosecant recurrence 1b with $c\to a\,c,\ d\to b\,c+a\,d,\ C\to b\,d,\ m\to 0,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\land\,m>1\,\land\,a^2-b^2\neq 0$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{b\,d\,\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-1}}{f\,m}+$$
$$\frac{1}{m}\int (a+b\,\mathrm{Sec}[e+f\,x])^{m-2}\,\left(a^2\,c\,m+\left(b^2\,d\,(m-1)+2\,a\,b\,c\,m+a^2\,d\,m\right)\,\mathrm{Sec}[e+f\,x]+b\,(b\,c\,m+a\,d\,(2\,m-1))\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)/(f*m) +
  1/m*Int[(a+b*Csc[e+f*x])^(m-2)*
    Simp[a^2*c*m+(b^2*d*(m-1)+2*a*b*c*m+a^2*d*m)*Csc[e+f*x]+b*(b*c*m+a*d*(2*m-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && GtQ[m,1] && NeQ[a^2-b^2,0] && IntegerQ[2*m]
```

2. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\land\,m<0$

1: $\int \frac{c+d\,\mathrm{Sec}[e+f\,x]}{a+b\,\mathrm{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{c+d\,z}{a+b\,z}==\frac{c}{a}-\frac{(b\,c-a\,d)\,z}{a\,(a+b\,z)}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{c + d \operatorname{Sec}[e + f x]}{a + b \operatorname{Sec}[e + f x]} \, dx \rightarrow \frac{c x}{a} - \frac{b c - a d}{a} \int \frac{\operatorname{Sec}[e + f x]}{a + b \operatorname{Sec}[e + f x]} \, dx$$

Program code:

```
Int[(c_+d_.*csc[e_.+f_.*x_])/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  c*x/a - (b*c-a*d)/a*Int[Csc[e+f*x]/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

2. $\int \frac{c + d \operatorname{Sec}[e + f x]}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx$ when $b c - a d \neq 0$

1: $\int \frac{c + d \operatorname{Sec}[e + f x]}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 == 0$

Derivation: Algebraic expansion

Basis: $\frac{c+d z}{\sqrt{a+b z}} == \frac{c \sqrt{a+b z}}{a} - \frac{(b c-a d) z}{a \sqrt{a+b z}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 == 0$, then

$$\int \frac{c + d \operatorname{Sec}[e + f x]}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx \rightarrow \frac{c}{a} \int \sqrt{a + b \operatorname{Sec}[e + f x]} \, dx - \frac{b c - a d}{a} \int \frac{\operatorname{Sec}[e + f x]}{\sqrt{a + b \operatorname{Sec}[e + f x]}} \, dx$$

Program code:

```
Int[(c_+d_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  c/a*Int[Sqrt[a+b*Csc[e+f*x]],x] - (b*c-a*d)/a*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0]
```

2: $\int \frac{c+d\,\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2\neq 0$

Derivation: Algebraic expansion

Rule: If $b\,c-a\,d\neq 0\land a^2-b^2\neq 0$, then

$$\int \frac{c+d\,\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx\;\rightarrow\;c\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+d\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(c_+d_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  c*Int[1/Sqrt[a+b*Csc[e+f*x]],x] + d*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

3. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge m<-1$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge m<-1\wedge a^2-b^2==0$

Derivation: Singly degenerate secant recurrence 2b with $n\to 0,\ p\to 0$

Rule: If $b\,c-a\,d\neq 0\wedge m<-1\wedge a^2-b^2==0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$\frac{(b\,c-a\,d)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m}{b\,f\,(2\,m+1)}+$$
$$\frac{1}{a^2\,(2\,m+1)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\,(a\,c\,(2\,m+1)-(b\,c-a\,d)\,(m+1)\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(b*f*(2*m+1)) +
  1/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*Simp[a*c*(2*m+1)-(b*c-a*d)*(m+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && EqQ[a^2-b^2,0] && IntegerQ[2*m]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge m<-1\wedge a^2-b^2\neq 0$

Derivation: Cosecant recurrence 2b with $C\to 0,\ m\to 0$

Rule: If $b\,c-a\,d\neq 0\wedge m<-1\wedge a^2-b^2\neq 0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$
$$-\frac{b\,(b\,c-a\,d)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}}{a\,f\,(m+1)\,(a^2-b^2)}+$$

$$\frac{1}{a(m+1)(a^2-b^2)}\int (a+b\,\text{Sec}[e+f x])^{m+1}\left(c(a^2-b^2)(m+1)-a(b c-a d)(m+1)\,\text{Sec}[e+f x]+b(b c-a d)(m+2)\,\text{Sec}[e+f x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  b*(b*c-a*d)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*
    Simp[c*(a^2-b^2)*(m+1)-(a*(b*c-a*d)*(m+1))*Csc[e+f*x]+b*(b*c-a*d)*(m+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && NeQ[a^2-b^2,0] && IntegerQ[2*m]
```

3: $\int (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])\,dx$ when $b c-a d\neq 0 \land 2m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $b c-a d\neq 0 \land 2m\notin\mathbb{Z}$, then

$$\int (a+b\,\text{Sec}[e+f x])^m (c+d\,\text{Sec}[e+f x])\,dx \to c\int (a+b\,\text{Sec}[e+f x])^m\,dx+d\int (a+b\,\text{Sec}[e+f x])^m\,\text{Sec}[e+f x]\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  c*Int[(a+b*Csc[e+f*x])^m,x] + d*Int[(a+b*Csc[e+f*x])^m*Csc[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[2*m]]
```

3. $\int \frac{(a+b\,\text{Sec}[e+f x])^m}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0$

1. $\int \frac{\sqrt{a+b\,\text{Sec}[e+f x]}}{c+d\,\text{Sec}[e+f x]}\,dx$ when $b c-a d\neq 0$

1: $\int \frac{\sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$ when $b c-a d \neq 0 \wedge \left(a^2-b^2==0 \vee c^2-d^2==0\right)$

Derivation: Algebraic expansion

Basis: $\frac{1}{c+d z} == \frac{1}{c} - \frac{d z}{c (c+d z)}$

Rule: If $b c-a d \neq 0 \wedge \left(a^2-b^2==0 \vee c^2-d^2==0\right)$, then

$$\int \frac{\sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx \rightarrow \frac{1}{c} \int \sqrt{a+b \operatorname{Sec}[e+f x]} \, dx - \frac{d}{c} \int \frac{\operatorname{Sec}[e+f x] \sqrt{a+b \operatorname{Sec}[e+f x]}}{c+d \operatorname{Sec}[e+f x]} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  1/c*Int[Sqrt[a+b*Csc[e+f*x]],x] - d/c*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ a^2-b^2\neq 0\ \wedge\ c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b\,z}}{c+d\,z} == \frac{a}{c\,\sqrt{a+b\,z}}+\frac{(b\,c-a\,d)\,z}{c\,\sqrt{a+b\,z}\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\ \wedge\ a^2-b^2\neq 0\ \wedge\ c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{a}{c}\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+\frac{b\,c-a\,d}{c}\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a/c*Int[1/Sqrt[a+b*Csc[e+f*x]],x] + (b*c-a*d)/c*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ (a^2-b^2==0\ \vee\ c^2-d^2==0)$

Derivation: Algebraic expansion

Basis: $\frac{(a+b\,z)^{3/2}}{c+d\,z} == \frac{a\,\sqrt{a+b\,z}}{c}+\frac{(b\,c-a\,d)\,z\,\sqrt{a+b\,z}}{c\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\ \wedge\ (a^2-b^2==0\ \vee\ c^2-d^2==0)$, then

$$\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{a}{c}\int\sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx+\frac{b\,c-a\,d}{c}\int\frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a/c*Int[Sqrt[a+b*Csc[e+f*x]],x] + (b*c-a*d)/c*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

x: $\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b\,z)^{3/2}}{c+d\,z}=\frac{b\sqrt{a+b\,z}}{d}-\frac{(b\,c-a\,d)\sqrt{a+b\,z}}{d\,(c+d\,z)}$

Note: This rule results in 3 `EllipticPi` terms.

Rule: If $b\,c-a\,d\neq 0\land a^2-b^2\neq 0\land c^2-d^2\neq 0$, then

$$\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{b}{d}\int\sqrt{a+b\,\text{Sec}[e+f\,x]}\,dx-\frac{b\,c-a\,d}{d}\int\frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$$

Program code:

```
(* Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  b/d*Int[Sqrt[a+b*Csc[e+f*x]],x] - (b*c-a*d)/d*Int[Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] *)
```

2: $\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

- Derivation: Algebraic expansion

- Basis: $\frac{(a+b\,z)^{3/2}}{c+d\,z}==\frac{(a+b\,z)^2}{\sqrt{a+b\,z}\,(c+d\,z)}==\frac{a^2\,d+b^2\,c\,z}{c\,d\,\sqrt{a+b\,z}}-\frac{(b\,c-a\,d)^2\,z}{c\,d\,\sqrt{a+b\,z}\,(c+d\,z)}$

- Note: This rule results in 2 `EllipticPi` terms and 1 `EllipticF` term.

- Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$, then

$$\int \frac{(a+b\,\text{Sec}[e+f\,x])^{3/2}}{c+d\,\text{Sec}[e+f\,x]}\,dx\,\rightarrow\,\frac{1}{c\,d}\int \frac{a^2\,d+b^2\,c\,\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx-\frac{(b\,c-a\,d)^2}{c\,d}\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,(c+d\,\text{Sec}[e+f\,x])}\,dx$$

- Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^(3/2)/(c_+d_.*csc[e_.+f_.*x_]),x_Symbol] :=
  1/(c*d)*Int[(a^2*d+b^2*c*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] -
  (b*c-a*d)^2/(c*d)*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3. $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,(a^2-b^2==0\,\vee\,c^2-d^2==0)$

- Derivation: Algebraic expansion

- Basis: $\frac{1}{\sqrt{a+b\,z}\;(c+d\,z)}==\frac{b\,c-a\,d-b\,d\,z}{c\,(b\,c-a\,d)\,\sqrt{a+b\,z}}+\frac{d^2\,z\,\sqrt{a+b\,z}}{c\,(b\,c-a\,d)\,(c+d\,z)}$

- Rule: If $b\,c-a\,d\neq 0\,\wedge\,(a^2-b^2==0\,\vee\,c^2-d^2==0)$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;(c+d\,\text{Sec}[e+f\,x])}\,dx\,\rightarrow\,\frac{1}{c\,(b\,c-a\,d)}\int\frac{b\,c-a\,d-b\,d\,\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+\frac{d^2}{c\,(b\,c-a\,d)}\int\frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{c+d\,\text{Sec}[e+f\,x]}\,dx$$

- Program code:

```
Int[1/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  1/(c*(b*c-a*d))*Int[(b*c-a*d-b*d*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] +
  d^2/(c*(b*c-a*d))*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && (EqQ[a^2-b^2,0] || EqQ[c^2-d^2,0])
```

2: $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;(c+d\,\text{Sec}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

- Derivation: Algebraic expansion

- Basis: $\frac{1}{c+d\,\text{Sec}[z]}==\frac{1}{c}-\frac{d}{c\,(d+c\,\text{Cos}[z])}$

- Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\,(c+d\,\text{Sec}[e+f\,x])}\,dx \to \frac{1}{c}\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx - \frac{d}{c}\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\,(c+d\,\text{Sec}[e+f\,x])}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*(c_+d_.*csc[e_.+f_.*x_])),x_Symbol] :=
  1/c*Int[1/Sqrt[a+b*Csc[e+f*x]],x] - d/c*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*(c+d*Csc[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0]
```

4. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(c+d\,\text{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m^2 == n^2 == \frac{1}{4}$

1. $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2 == 0\,\wedge\,c^2-d^2 == 0$

Derivation: Piecewise constant extraction

Basis: If $a^2-b^2 == 0\,\wedge\,c^2-d^2 == 0$, then $\partial_x \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}{\text{Tan}[e+f\,x]} == 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2 == 0\,\wedge\,c^2-d^2 == 0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}\,dx \to \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}{\text{Tan}[e+f\,x]}\int \text{Tan}[e+f\,x]\,dx$$

–

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]/Cot[e+f*x]*Int[Cot[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{c+d\,z} == \frac{c}{\sqrt{c+d\,z}}+\frac{d\,z}{\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}\,dx \to c\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx+d\int \frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  c*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] +
  d*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

2. $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0$

1. $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0$

1: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2-b^2==0\wedge c^2-d^2==0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{c+d z}} == \frac{\sqrt{c+d z}}{c} - \frac{d z}{c \sqrt{c+d z}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 == 0$, then

$$\int \frac{\sqrt{a + b \operatorname{Sec}[e + f x]}}{\sqrt{c + d \operatorname{Sec}[e + f x]}} \, dx \to \frac{1}{c} \int \sqrt{a + b \operatorname{Sec}[e + f x]} \sqrt{c + d \operatorname{Sec}[e + f x]} \, dx - \frac{d}{c} \int \frac{\operatorname{Sec}[e + f x] \sqrt{a + b \operatorname{Sec}[e + f x]}}{\sqrt{c + d \operatorname{Sec}[e + f x]}} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  1/c*Int[Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]],x] -
  d/c*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2=0\,\land\,c^2-d^2\neq 0$

Derivation: Integration by substitution

Basis: If $a^2-b^2=0$, then

$$\frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}=\frac{2\,a}{f}\,\text{Subst}\left[\frac{1}{1+a\,c\,x^2},\,x,\,\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\right]\partial_x\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}$$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2=0\,\land\,c^2-d^2\neq 0$, then

$$\int\frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx\,\to\,\frac{2\,a}{f}\,\text{Subst}\left[\int\frac{1}{1+a\,c\,x^2}\,dx,\,x,\,\frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  -2*a/f*Subst[Int[1/(1+a*c*x^2),x],x,Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2. $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0$

1: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2==0$

Derivation: Algebraic expansion

Basis: $\frac{\sqrt{a+b\,z}}{\sqrt{c+d\,z}}==\frac{a\,\sqrt{c+d\,z}}{c\,\sqrt{a+b\,z}}+\frac{(b\,c-a\,d)\,z}{c\,\sqrt{a+b\,z}\,\sqrt{c+d\,z}}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2==0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx\,\rightarrow\,\frac{a}{c}\int \frac{\sqrt{c+d\,\text{Sec}[e+f\,x]}}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+\frac{b\,c-a\,d}{c}\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  a/c*Int[Sqrt[c+d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  (b*c-a*d)/c*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx\,\rightarrow$$

$$-\frac{2\,(a+b\,\mathrm{Sec}[e+f\,x])}{c\,f\,\sqrt{\frac{a+b}{c+d}}\,\mathrm{Tan}[e+f\,x]}\sqrt{\frac{(b\,c-a\,d)\,(1+\mathrm{Sec}[e+f\,x])}{(c-d)\,(a+b\,\mathrm{Sec}[e+f\,x])}}$$

$$\sqrt{-\frac{(b\,c-a\,d)\,(1-\mathrm{Sec}[e+f\,x])}{(c+d)\,(a+b\,\mathrm{Sec}[e+f\,x])}}\;\mathrm{EllipticPi}\Big[\frac{a\,(c+d)}{c\,(a+b)},\,\mathrm{ArcSin}\Big[\sqrt{\frac{a+b}{c+d}}\,\frac{\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}\Big],\,\frac{(a-b)\,(c+d)}{(a+b)\,(c-d)}\Big]$$

–

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/Sqrt[c_+d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  2*(a+b*Csc[e+f*x])/(c*f*Rt[(a+b)/(c+d),2]*Cot[e+f*x])*
    Sqrt[(b*c-a*d)*(1+Csc[e+f*x])/((c-d)*(a+b*Csc[e+f*x]))]*
    Sqrt[-(b*c-a*d)*(1-Csc[e+f*x])/((c+d)*(a+b*Csc[e+f*x]))]*
    EllipticPi[a*(c+d)/(c*(a+b)),ArcSin[Rt[(a+b)/(c+d),2]*Sqrt[c+d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]]],(a-b)*(c+d)/((a+b)*(c-d))] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

3. $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2==0$

Derivation: Piecewise constant extraction

Basis: If $a^2-b^2==0\,\wedge\,c^2-d^2==0$, then $\partial_x \frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}==0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2==0\,\wedge\,c^2-d^2==0$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx \to \frac{\text{Tan}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\int \frac{1}{\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  Cot[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*Int[1/Cot[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && EqQ[c^2-d^2,0]
```

2: $\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\,\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{a+b\,z}}==\frac{1}{a}\sqrt{a+b\,z}-\frac{b\,z}{a\,\sqrt{a+b\,z}}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{1}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx\;\to\;\frac{1}{a}\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx-\frac{b}{a}\int \frac{\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}\;\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*csc[e_.+f_.*x_]]*Sqrt[c_+d_.*csc[e_.+f_.*x_]]),x_Symbol] :=
  1/a*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] -
  b/a*Int[Csc[e+f*x]/(Sqrt[a+b*Csc[e+f*x]]*Sqrt[c+d*Csc[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0]
```

5: $\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{(c+d\,\text{Sec}[e+f\,x])^{3/2}}\,dx$ when $b\,c-a\,d\neq 0\;\land\;c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{c+d\,z}==\frac{1}{c}-\frac{d\,z}{c\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\;\land\;c^2-d^2\neq 0$, then

$$\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{(c+d\,\text{Sec}[e+f\,x])^{3/2}}\,dx\;\to\;\frac{1}{c}\int \frac{\sqrt{a+b\,\text{Sec}[e+f\,x]}}{\sqrt{c+d\,\text{Sec}[e+f\,x]}}\,dx-\frac{d}{c}\int \frac{\text{Sec}[e+f\,x]\,\sqrt{a+b\,\text{Sec}[e+f\,x]}}{(c+d\,\text{Sec}[e+f\,x])^{3/2}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_.+f_.*x_]]/(c_+d_.*csc[e_.+f_.*x_])^(3/2),x_Symbol] :=
  1/c*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[c+d*Csc[e+f*x]],x] -
  d/c*Int[Csc[e+f*x]*Sqrt[a+b*Csc[e+f*x]]/(c+d*Csc[e+f*x])^(3/2),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[c^2-d^2,0]
```

6: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m-\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2==0$, then $\partial_x \frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}==0$

Basis: If $a^2-b^2==0$, then $-\frac{a^2\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\,\frac{\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}==1$

Basis: $\mathrm{Tan}[e+f\,x]==\frac{1}{f}\,\mathrm{Subst}\left[\frac{1}{x},\,x,\,\mathrm{Sec}[e+f\,x]\right]\,\partial_x\mathrm{Sec}[e+f\,x]$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m-\frac{1}{2}\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{a^2\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\int\frac{\mathrm{Tan}[e+f\,x]\,(a+b\,\mathrm{Sec}[e+f\,x])^{m-\frac{1}{2}}\,(c+d\,\mathrm{Sec}[e+f\,x])^n}{\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\,dx\,\longrightarrow$$

$$-\frac{a^2\,\mathrm{Tan}[e+f\,x]}{f\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}\,\sqrt{a-b\,\mathrm{Sec}[e+f\,x]}}\,\mathrm{Subst}\left[\int\frac{(a+b\,x)^{m-\frac{1}{2}}\,(c+d\,x)^n}{x\,\sqrt{a-b\,x}}\,dx,\,x,\,\mathrm{Sec}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^2*Cot[e+f*x]/(f*Sqrt[a+b*Csc[e+f*x]]*Sqrt[a-b*Csc[e+f*x]])*
    Subst[Int[(a+b*x)^(m-1/2)*(c+d*x)^n/(x*Sqrt[a-b*x]),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && IntegerQ[m-1/2]
```

7. $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n\in\mathbb{Z}$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $m+n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, then $(a+b\,\mathrm{Sec}[z])^m\,(c+d\,\mathrm{Sec}[z])^n == \frac{(b+a\,\mathrm{Cos}[z])^m\,(d+c\,\mathrm{Cos}[z])^n}{\mathrm{Cos}[z]^{m+n}}$

Note: The restriction $m+n\in\{0,\,-1,\,-2\}$ can be lifted if and when the cosine integration rules are extended to handle integrands of the form $\cos[e+f\,x]^p\,(a+b\,\cos[e+f\,x])^m\,(c+d\,\cos[e+f\,x])^n$ for arbitray p.

Rule: If $b\,c-a\,d\neq 0\,\land\,m\in\mathbb{Z}\,\land\,n\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\longrightarrow\,\int \frac{(b+a\,\mathrm{Cos}[e+f\,x])^m\,(d+c\,\mathrm{Cos}[e+f\,x])^n}{\mathrm{Cos}[e+f\,x]^{m+n}}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^(m+n),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && IntegerQ[m] && IntegerQ[n] && LeQ[-2,m+n,0]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m+\frac{1}{2}\in\mathbb{Z}\,\wedge\,n+\frac{1}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\sqrt{d+c\,\mathrm{Cos}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{\sqrt{b+a\,\mathrm{Cos}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}==0$

Note: The restriction $m+n\in\{0,-1,-2\}$ can be lifted if and when the cosine integration rules are extended to handle integrands of the form $\mathrm{Cos}[e+f\,x]^p\,(a+b\,\mathrm{Cos}[e+f\,x])^m\,(c+d\,\mathrm{Cos}[e+f\,x])^n$ for arbitray p.

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m+\frac{1}{2}\in\mathbb{Z}\,\wedge\,n+\frac{1}{2}\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\,\to\,\frac{\sqrt{d+c\,\mathrm{Cos}[e+f\,x]}\,\sqrt{a+b\,\mathrm{Sec}[e+f\,x]}}{\sqrt{b+a\,\mathrm{Cos}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Sec}[e+f\,x]}}\int\frac{(b+a\,\mathrm{Cos}[e+f\,x])^m\,(d+c\,\mathrm{Cos}[e+f\,x])^n}{\mathrm{Cos}[e+f\,x]^{m+n}}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Sqrt[d+c*Sin[e+f*x]]*Sqrt[a+b*Csc[e+f*x]]/(Sqrt[b+a*Sin[e+f*x]]*Sqrt[c+d*Csc[e+f*x]])*
    Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^(m+n),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && IntegerQ[m+1/2] && IntegerQ[n+1/2] && LeQ[-2,m+n,0]
```

3: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m+n==0\,\wedge\,2\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\mathrm{Cos}[e+f\,x]^{m+n}\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n}{(b+a\,\mathrm{Cos}[e+f\,x])^m\,(d+c\,\mathrm{Cos}[e+f\,x])^n}==0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m+n==0\,\wedge\,2\,m\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx \;\longrightarrow\; \frac{\mathrm{Cos}[e+f\,x]^{m+n}\,(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n}{(b+a\,\mathrm{Cos}[e+f\,x])^m\,(d+c\,\mathrm{Cos}[e+f\,x])^n}\int \frac{(b+a\,\mathrm{Cos}[e+f\,x])^m\,(d+c\,\mathrm{Cos}[e+f\,x])^n}{\mathrm{Cos}[e+f\,x]^{m+n}}\,dx$$

–

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Sin[e+f*x]^(m+n)*(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n/((b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n)*
    Int[(b+a*Sin[e+f*x])^m*(d+c*Sin[e+f*x])^n/Sin[e+f*x]^Simplify[m+n],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && EqQ[m+n,0] && Not[IntegerQ[2*m]]
```

8: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$ when $n \in \mathbb{Z}^+$

–

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx \;\longrightarrow\; \int (a+b\,\mathrm{Sec}[e+f\,x])^m\,\mathrm{ExpandTrig}\left[(c+d\,\mathrm{Sec}[e+f\,x])^n,\,x\right]dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(c_+d_.*csc[e_.+f_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(a+b*csc[e+f*x])^m,(c+d*csc[e+f*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[n,0]
```

X: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$

Rule:

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx\;\longrightarrow\;\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c+d\,\mathrm{Sec}[e+f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(c_+d_.*csc[e_.+f_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(c+d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n},x]
```

Rules for integrands of the form $(a+b\,\mathrm{Sec}[e+f\,x])^m\,(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n$

1: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Cos}[e+f\,x])^n\,dx$ when $n\notin\mathbb{Z}\land m\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $m\in\mathbb{Z}$, then $(a+b\,\mathrm{Sec}[z])^m == \frac{d^m\,(b+a\,\mathrm{Cos}[z])^m}{(d\,\mathrm{Cos}[z])^m}$

Rule: If $n\notin\mathbb{Z}\land m\in\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Cos}[e+f\,x])^n\,dx \longrightarrow d^m\int (b+a\,\mathrm{Cos}[e+f\,x])^m\,(d\,\mathrm{Cos}[e+f\,x])^{n-m}\,dx$$

Program code:

```
Int[(a_.+b_.*sec[e_.+f_.*x_])^m_.*(d_./sec[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(b+a*Cos[e+f*x])^m*(d*Cos[e+f*x])^(n-m),x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_.+b_.*csc[e_.+f_.*x_])^m_.*(d_./csc[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(b+a*Sin[e+f*x])^m*(d*Sin[e+f*x])^(n-m),x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

2: $\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,(d\,\mathrm{Sec}[e+f\,x])^p)^n}{(d\,\mathrm{Sec}[e+f\,x])^{n\,p}} == 0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,\left(c\,(d\,\mathrm{Sec}[e+f\,x])^p\right)^n\,dx \longrightarrow \frac{c^{\mathrm{IntPart}[n]}\,\left(c\,(d\,\mathrm{Sec}[e+f\,x])^p\right)^{\mathrm{FracPart}[n]}}{(d\,\mathrm{Sec}[e+f\,x])^{p\,\mathrm{FracPart}[n]}}\int (a+b\,\mathrm{Sec}[e+f\,x])^m\,(d\,\mathrm{Sec}[e+f\,x])^{n\,p}\,dx$$

– Program code:

```
Int[(a_.+b_.*sec[e_.+f_.*x_])^m_.*(c_.*(d_.*sec[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Sec[e + f*x])^p)^FracPart[n]/(d*Sec[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Sec[e+f*x])^m*(d*Sec[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]]
```

```
Int[(a_.+b_.*csc[e_.+f_.*x_])^m_.*(c_.*(d_.*csc[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Csc[e + f*x])^p)^FracPart[n]/(d*Csc[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Cos[e+f*x])^m*(d*Cos[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]]
```